get their input until they press enter
,----------[++++++++++>,----------]
go to beginning of input
<[<]>
subtract 47 from all of their input (ASCII "0" is 48 and ASCII "1" is 49 and so
on so this turns the character "1" into 0x2 and the character "4" into 0x5 and
so on) (We subtract 47 so that "0" doesn't become 0x0 which would end the loops)
[----------------------------------------------->]
go to beginning of numbers
<[<]>
add 10*(current pointer value minus one) to next value over as long as there are
next valus over
>[<-[->++++++++++<]>>]
dec and output final answer
<-.
clear cell
[-]
print newline
++++++++++.
made by chris lutz
